

# Jane Eslinger

Principal Technical Support Analyst at SAS

Jane has authored two books: her latest is The SAS® Programmer's PROC REPORT Handbook: ODS Companion.

# It's All About the Base—Procedures

Jane Eslinger, SAS Institute Inc.

# Procedures

UTILITY	STATISTICAL-ESQUE	REPORTING
Append	Freq	Print
Compare	Means and Summary	Report
Contents	Univariate	Tabulate
Datasets		
Format		
Sort		
SQL		
Transpose		

# APPEND

adds observations in one data set to bottom of another data set

# APPEND

## Challenges

1. Data structure of the two data sets must match
2. Observations from either data set cannot be modified

# APPEND

## Best Use Case

Add the latest data to the historical data:

```
proc append base=prdsal2 data=sale2000;  
run;
```

# APPEND

## Comparison

- DATA Step:

```
data allsale;  
    set prdsal2(rename=(predict=orig_predict))  
        sale2000(rename=(predict=orig_predict));  
  
    /*adjust prediction value*/  
    predict = orig_predict * 1.01;  
run;
```

# COMPARE

compares the variable values and attributes of two data sets



# COMPARE

## Challenges

1. Does not output values that are the same
2. Does not provide list of BY or ID values that are in one data set but not the other, or in both
3. Lacks options for altering the formatting in the report
4. Difficult to examine many differences

# COMPARE

## Best Use Case

Find the one or two variables that might have differences:

```
proc compare data=sashelp.prdsal2  
             comp=allsale noprint outdif  
             outcomp outnoequal outbase  
             out=diffs;  
  
run;
```

# COMPARE

## Comparison

- Not Applicable

# CONTENTS

provides data set attribute information, engine and host information, lists variables

# CONTENTS

## Challenges

1. Number of observations for a DATA step view not reported
2. **Created By** and **Last Modified By** characteristics are unique to Z/OS sequential access bound libraries
3. Number of observations that are contained in an external database not provided

# CONTENTS

## Best Use Case

Check variables in data set:

```
proc contents data=sashelp.prdsal2  
              out=contents;  
  
run;
```

# CONTENTS

## Comparisons

- PROC DATASETS
- PROC SQL

# DATASETS

changes variable attributes, renames variables, creates indexes,  
displays contents of a data set



# DATASETS

## Challenges

1. Will not drop variables
2. Will not provide the number of observations that are contained in an external database table

# DATASETS

## Best Use Case

Modify formats and labels:

```
proc datasets library=work nolist;  
    modify prdsal2;  
    label product='Product Name';  
    format country $char20.;  
quit;
```

# DATASETS

## Comparisons

- PROC CONTENTS
- PROC SQL
- DATA Step

# FORMAT

creates informats, formats, and picture formats

# FORMAT

## Challenges

1. You must know how you want your data values displayed

# FORMAT

## Best Use Case

Group values into categories:

```
proc format;  
  value $func  
    'BED', 'SOFA' = 'Soft'  
    'DESK', 'CHAIR' = 'Hard';  
run;
```

# FORMAT

## Comparison

- Not Applicable



<https://tenor.com/view/tired-sleepy-exhausted-working-work-gif-7995639>



# SORT

sorts data,  
removes duplicate observations

# SORT

## Challenges

1. Amount of memory and disk space that is used can be problematic for large data sets

# SORT

## Best Use Case

Sort data:

```
proc sort data=sashelp.prdsal2  
          out=prdsal2;  
    by year product country;  
run;
```

# SORT

## Comparison

- PROC SQL

```
proc sql noprint;  
    create table unqprd2 as  
    select distinct country, state, product  
    from sashelp.prdsal2  
    order by country, state, product;  
quit;
```

# SQL

sorts, merges, creates indexes,  
renames variables, calculates statistics, and more

# SQL

## Challenges

1. Is not standard SAS procedure with rigid syntax – a huge learning curve
2. Does not create production-quality report

# SQL

## Best Use Case

Create a macro variable with a list of values:

```
proc sql noprint;  
    select distinct(state) into :stlist1  
        separated by ', '  
    from sashelp.prdsal2;  
quit;
```

# SQL

## Comparisons

- PROC DATASETS/PROC CONTENTS
- PROC FREQ
- PROC MEANS
- PROC SORT
- DATA Step



# TRANSPOSE

restructures data

# TRANSPOSE

## Challenges

1. Requires sorting first in most cases
2. Controlling the resulting variable names

# TRANSPOSE

## Best Use Case

Create new numeric variables named after the value of a categorical variable:

```
proc transpose data=sashelp.stocks  
               out=test;  
  where year(date)=2005;  
  by stock;  
  id date;  
  var close;  
run;
```

# TRANSPOSE

## Comparison

- DATA Step

# FREQ

provides counts and percentages,  
generates statistics

# FREQ

## Challenges

1. Categories that are not present in the data are not displayed
2. OUT= and ODS output data sets do not contain the same variables or structure
3. Format of the statistics might be difficult to change
4. Customized percentages are not possible

# FREQ

## Best Use Case

Check data to ensure it contains all expected combinations:

```
proc freq data=sashelp.prdsal2;  
    tables country*product /list out=freqs;  
run;
```

# FREQ

## The FREQ Procedure

COUNTRY	PRODUCT	Frequency	Percent	Cumulative Frequency	Cumulative Percent
Canada	BED	1152	5.00	1152	5.00
Canada	CHAIR	1152	5.00	2304	10.00
Canada	DESK	1152	5.00	3456	15.00
Canada	SOFA	1152	5.00	4608	20.00
Mexico	BED	1152	5.00	5760	25.00
Mexico	CHAIR	1152	5.00	6912	30.00
Mexico	DESK	1152	5.00	8064	35.00
Mexico	SOFA	1152	5.00	9216	40.00
U.S.A.	BED	3456	15.00	12672	55.00
U.S.A.	CHAIR	3456	15.00	16128	70.00
U.S.A.	DESK	3456	15.00	19584	85.00
U.S.A.	SOFA	3456	15.00	23040	100.00



# FREQ

## Comparisons

- PROC REPORT
- PROC TABULATE

# MEANS and SUMMARY

calculates statistics and percentiles

# MEANS and SUMMARY

## Challenges

1. Do not calculate percentages
2. Structure of report output is either difficult or impossible to change

# MEANS and SUMMARY

## Best Use Case

Create data sets with statistics, totals, and overall means:

```
proc means data=sashelp.prdsal2
            completetypes nway;
  format product $prod.;
  class product / preloadfmt;
  var actual predict;
  output out=means mean= sum= n=/
         autoname;

run;
```

# MEANS and SUMMARY

	PRODUCT	_TYPE_	_FREQ_	ACTUAL_Mean	PREDICT_Mean	ACTUAL_Sum	PREDICT_Sum	ACTUAL_N	PREDICT_N
1	BED	1	5760	\$644.14	\$678.73	\$3710251.84	\$3909496.40	5760	5760
2	CHAIR	1	5760	\$641.56	\$694.13	\$3695383.62	\$3998203.09	5760	5760
3	DESK	1	5760	\$645.55	\$696.09	\$3718359.00	\$4009475.23	5760	5760
4	SOFA	1	5760	\$674.34	\$695.45	\$3884187.86	\$4005805.83	5760	5760
5	TABLE	1	0	.	.	.	.	0	0

# MEANS and SUMMARY

The MEANS Procedure

Product	N Obs	Variable	Label	N	Mean	Std Dev	Minimum	Maximum
BED	5760	ACTUAL	Actual Sales	5760	644.1409447	654.5324375	0.0615385	3440.80
		PREDICT	Predicted Sales	5760	678.7320144	672.2739927	0	3605.70
CHAIR	5760	ACTUAL	Actual Sales	5760	641.5596563	640.5110277	0	3376.20
		PREDICT	Predicted Sales	5760	694.1324808	691.6989891	0	3564.90
DESK	5760	ACTUAL	Actual Sales	5760	645.5484375	656.2886439	0	3415.30
		PREDICT	Predicted Sales	5760	696.0894495	687.8018861	0	3609.10
SOFA	5760	ACTUAL	Actual Sales	5760	674.3381707	673.9880104	0	3515.60
		PREDICT	Predicted Sales	5760	695.4524014	688.4784458	0	3634.60
TABLE	0	ACTUAL	Actual Sales	0				
		PREDICT	Predicted Sales	0				

# MEANS and SUMMARY

## Comparisons

- PROC SQL
- PROC TABULATE

# UNIVARIATE

calculates summary statistics, generates quantiles and percentiles, performs tests for goodness of fit, and produces probability plots



# UNIVARIATE

## Challenges

1. Default output contains a large amount of tables and data
2. No options for subsetting statistic list

# UNIVARIATE

## Best Use Case

Generate custom (non-default) percentiles:

```
proc univariate data=sashelp.prdsal2;  
  var actual;  
  output out=univ pctlpts=15 25 35  
    pctlpre=actual_p;  
run;
```

# UNIVARIATE

## Comparison

- PROC MEANS and PROC SUMMARY

```
proc means data=sashelp.prdsal2 min max  
          mean mode;  
  class product;  
  var predict actual;  
  output out=stats mean= sum= /autoname;  
run;
```



# PRINT

displays data,  
calculates summary totals for numeric data

# PRINT

## Challenges

1. Customization is limited
2. Numeric values cannot be added, subtracted, multiplied, or divided against each other
3. Output data set cannot be created

# PRINT

## Best Use Case

Check data to ensure it has expected values:

```
proc print data=sashelp.prdsal2 (obs=30) ;  
    var country state product actual  
        predict;  
run;
```

# PRINT

Obs	COUNTRY	STATE	PRODUCT	ACTUAL	PREDICT
1	U.S.A.	California	SOFA	\$987.36	\$692.24
2	U.S.A.	California	SOFA	\$1,782.96	\$568.48
3	U.S.A.	California	SOFA	\$32.64	\$16.32
4	U.S.A.	California	SOFA	\$1,825.12	\$756.16
5	U.S.A.	California	SOFA	\$750.72	\$723.52
6	U.S.A.	California	SOFA	\$2,426.24	\$2,428.96
7	U.S.A.	California	SOFA	\$1,791.12	\$2,250.80
8	U.S.A.	California	SOFA	\$2,282.08	\$350.88
9	U.S.A.	California	SOFA	\$2,518.72	\$1,736.72
10	U.S.A.	California	SOFA	\$1,436.16	\$2,167.84
11	U.S.A.	California	SOFA	\$2,314.72	\$62.56
12	U.S.A.	California	SOFA	\$1,410.32	\$1,670.08
13	U.S.A.	California	BED	\$369.92	\$1,365.44
14	U.S.A.	California	BED	\$2,014.16	\$2,358.24
15	U.S.A.	California	BED	\$85.68	\$2,594.88



# PRINT

## Comparison

- PROC REPORT

# REPORT

generates both detailed and summary reports

# REPORT

## Challenges

1. A learning curve to using the compute block
2. Does not provide a subtotals breakdown

# REPORT

## Best Use Case

### Generate final report:

```
proc report data=sashelp.prdsal2;  
  column country state product predict actual  
         diff;  
  define country / group noprint;  
  define state / group;  
  define product / group;  
  define diff / computed 'Dif=Pred-Act'  
                  format=dollar10.2;  
  break after country / summarize;
```

# REPORT

## Best Use Case (continued)

```
compute before country;  
    line 'Section for ' country $20.;  
endcomp;  
compute after country;  
    state = 'Total';  
endcomp;  
compute diff;  
    diff = predict.sum - actual.sum;  
endcomp;  
run;
```

# REPORT

State Province	Product	Predicted Sales	Actual Sales	Dif=Pred-Act
Section for Canada				
British Columbia	BED	\$225,369.00	\$197,706.60	\$27,662.40
	CHAIR	\$204,264.00	\$200,905.20	\$3,358.80
	DESK	\$194,682.60	\$186,262.20	\$8,420.40
	SOFA	\$230,587.20	\$216,282.60	\$14,304.60
Ontario	BED	\$210,796.20	\$194,493.60	\$16,302.60
	CHAIR	\$206,632.80	\$179,892.00	\$26,740.80
	DESK	\$206,857.80	\$208,778.40	\$-1,920.60
	SOFA	\$221,410.80	\$196,882.20	\$24,528.60
Quebec	BED	\$219,006.00	\$204,737.40	\$14,268.60
	CHAIR	\$225,808.20	\$199,751.40	\$26,056.80
	DESK	\$215,080.20	\$202,555.80	\$12,524.40
	SOFA	\$227,037.60	\$200,777.40	\$26,260.20
Saskatchewan	BED	\$195,429.60	\$193,568.40	\$1,861.20
	CHAIR	\$210,690.00	\$201,580.20	\$9,109.80
	DESK	\$228,223.80	\$208,481.40	\$19,742.40
	SOFA	\$209,140.20	\$205,178.40	\$3,961.80
<b>Total</b>		<b>\$3431016.00</b>	<b>\$3197833.20</b>	<b>\$233182.80</b>

# REPORT

## Comparison

- PROC PRINT
- PROC TABULATE

# TABULATE

calculates statistics and percentages,  
summarizes at a categorical level



# TABULATE

## Challenges

1. Syntax for the TABLE statement has a learning curve
2. Cannot perform basic mathematical calculations between columns
3. Formatting cannot be based on the value of another column
4. Cannot insert text in the middle of the table

# TABULATE

## Best Use Case

Generate percentages, calculates totals and subtotals:

```
proc tabulate data=sashelp.prdsal2;  
  class country product;  
  var actual;  
  table country*product  
         all='TOTALS'*product=' ',  
         actual*(sum*f=dollar12.  
         pctsum<product>);  
  
run;
```

# TABULATE

		Actual Sales	
		Sum	PctSum
Country	Product		
Canada	BED	\$790,506	24.72
	CHAIR	\$782,129	24.46
	DESK	\$806,078	25.21
	SOFA	\$819,121	25.61
Mexico	BED	\$535,408	25.37
	CHAIR	\$515,390	24.42
	DESK	\$523,075	24.78
	SOFA	\$536,870	25.44
U.S.A.	BED	\$2,384,338	24.58
	CHAIR	\$2,397,864	24.72
	DESK	\$2,389,207	24.63
	SOFA	\$2,528,197	26.06
TOTALS	BED	\$3,710,252	24.72
	CHAIR	\$3,695,384	24.62
	DESK	\$3,718,359	24.78
	SOFA	\$3,884,188	25.88

# TABULATE

## Comparison

- PROC REPORT



# Thank you!

Contact Information  
[support@sas.com](mailto:support@sas.com)

## Reminder:

Complete your session survey in the conference mobile app.